Function Definitions

In[1]:=

```
(*change in angle due to piezos*)
Δθ[x1_, x2_] = (x2 - x1)/q;
(*the location of the pivot point between the piezo actuators*)
Py[x1_, x2_] = (x2 + x1)/2;

(*the vector going from  the pivot point to the plane of
  the grating. It is perpendicular to the plane of the piezos.*)
ax[x1_, x2_] = a Cos[π/2 + Δθ[x1, x2]];
ay[x1_, x2_] = a Sin[π/2 + Δθ[x1, x2]];

(*a vector pointing along the direction of the plane of the grating*)
ξx[x1_, x2_, ξ_] = ξ * Cos[θ1 + Δθ[x1, x2]];
ξy[x1_, x2_, ξ_] = ξ * Sin[θ1 + Δθ[x1, x2]];

(*a vector representing where the laser hits the grating*)
Lx[x1_, x2_, ξ_] = ax[x1, x2] + ξx[x1, x2, ξ];
Ly[x1_, x2_, ξ_] = ay[x1, x2] + ξy[x1, x2, ξ] + Py[x1, x2];
```

Solve the equation

In[9]:=

```
(*solve. α represents how much  the cavity is shortened by. Essentially,
we say that the new place where the laser is striking the
  grating (Lx[x1,x2,ξ]) is equal to the old place (Lx[0,0,ξ0]) plus
  some  amount  α pointed in the direction of the laser beam *)
eq1 = Lx[x1, x2, ξ] == Lx[0, 0, ξ0] + α Cos[θL]
eq2 = Ly[x1, x2, ξ] == Ly[0, 0, ξ0] + α Sin[θL]
```

Out[9]= $\xi \operatorname{Cos}\left[\frac{-x1+x2}{q}+\theta 1\right]-a \operatorname{Sin}\left[\frac{-x1+x2}{q}\right]==\xi 0 \operatorname{Cos}[\theta 1]+\alpha \operatorname{Cos}[\theta L]$

Out[10]= $\frac{x1+x2}{2}+a \operatorname{Cos}\left[\frac{-x1+x2}{q}\right]+\xi \operatorname{Sin}\left[\frac{-x1+x2}{q}+\theta 1\right]==a+\xi 0 \operatorname{Sin}[\theta 1]+\alpha \operatorname{Sin}[\theta L]$

Define some convenient functions based on the solution

αFUNCT is the function representing how much shorter the cavity gets depending on x1 and x2. ξFUNCT tells you how far along the grating you need to go from your center point to get there. It is mostly just a maintenance function

In[11]:=
```
Solve[{eq1, eq2}, α, ξ];
αFUNCT[x1_, x2_] = %[[1, 1, 2]] // Simplify
Solve[{eq1, eq2}, ξ, α];
ξFUNCT[x1_, x2_] = %[[1, 1, 2]] // Simplify
```

Solve::bdomv : Warning: ξ is not a valid domain specification . Assuming it is a variable to eliminate . ≫

Out[12]= $\frac{1}{2} \mathrm{Csc}\left[\frac{x1 - x2 - q\,\theta 1 + q\,\theta L}{q}\right] \left(2\,a\,\mathrm{Cos}[\theta 1] + (-2\,a + x1 + x2)\,\mathrm{Cos}\left[\frac{x1 - x2 - q\,\theta 1}{q}\right] - 2\,\xi 0\,\mathrm{Sin}\left[\frac{x1 - x2}{q}\right]\right)$

Solve::bdomv : Warning: α is not a valid domain specification . Assuming it is a variable to eliminate . ≫

Out[14]= $\frac{1}{2} \mathrm{Csc}\left[\frac{x1 - x2 - q\,\theta 1 + q\,\theta L}{q}\right] \left((-2\,a + x1 + x2)\,\mathrm{Cos}[\theta L] + 2\,a\,\mathrm{Cos}\left[\frac{x1 - x2 + q\,\theta L}{q}\right] - 2\,\xi 0\,\mathrm{Sin}[\theta 1 - \theta L]\right)$

Now, use the condition that the change in λ due to the changing grating angle must match the change in λ due to the change in cavity length

In[15]:=
```
(*this is the change in λ due to the change in cavity
  length (recall αFUNCT is how much  SHORTER the cavity is*)
δλ1 = -(αFUNCT[x1, x2]/L) λ
(*this is the change in λ because of the change in the grating's angle*)
(*I believe the convention is that a larger Δθ corresponds to the
  grating angling away from  the light, meaning  it favors longer λs*)
δλ2 = 2 d Sin[gratingθ+ Δθ[x1, x2]] - 2 d Sin[gratingθ]
```

Out[15]= $-\frac{1}{2\,L} \lambda\,\mathrm{Csc}\left[\frac{x1 - x2 - q\,\theta 1 + q\,\theta L}{q}\right] \left(2\,a\,\mathrm{Cos}[\theta 1] + (-2\,a + x1 + x2)\,\mathrm{Cos}\left[\frac{x1 - x2 - q\,\theta 1}{q}\right] - 2\,\xi 0\,\mathrm{Sin}\left[\frac{x1 - x2}{q}\right]\right)$

Out[16]= $-2\,d\,\mathrm{Sin}[\mathrm{grating}\theta] + 2\,d\,\mathrm{Sin}\left[\mathrm{grating}\theta + \frac{-x1 + x2}{q}\right]$

Now, the change in λ due to both the changing angle and the changing cavity length should be the same. Thus, subtracting these two should give us a constant number.

In[17]:=
```
aBigConstant= δλ1 - δλ2
```

Out[17]= $2\,d\,\mathrm{Sin}[\mathrm{grating}\theta] - \frac{1}{2\,L} \lambda\,\mathrm{Csc}\left[\frac{x1 - x2 - q\,\theta 1 + q\,\theta L}{q}\right] \left(2\,a\,\mathrm{Cos}[\theta 1] + (-2\,a + x1 + x2)\,\mathrm{Cos}\left[\frac{x1 - x2 - q\,\theta 1}{q}\right] - 2\,\xi 0\,\mathrm{Sin}\left[\frac{x1 - x2}{q}\right]\right) - 2\,d\,\mathrm{Sin}\left[\mathrm{grating}\theta + \frac{-x1 + x2}{q}\right]$

In[18]:=
```
aBigConstant = aBigConstant //. x2 → myRATIO1 *x1;
D[aBigConstant, x1] //. x1 → 0;
Solve[% == 0, myRATIO1 ];
RATIO[λ_, a_, q_, θ1_, d_, L_, ξ0_] = %[[1, 1, 2]]
```

Out[21]=
$$\left(4\,d\,L\,\mathrm{Cos}[\mathrm{grating}\theta]+2\,\lambda\,\xi 0\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]-q\,\lambda\,\mathrm{Cos}[\theta 1]\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]+2\,a\,\lambda\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]\mathrm{Sin}[\theta 1]\right)\Big/\left(4\,d\,L\,\mathrm{Cos}[\mathrm{grating}\theta]+2\,\lambda\,\xi 0\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]+q\,\lambda\,\mathrm{Cos}[\theta 1]\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]+2\,a\,\lambda\,\mathrm{Csc}\left[\frac{-q\,\theta 1+q\,\theta L}{q}\right]\mathrm{Sin}[\theta 1]\right)$$

This is a symbolic expression for the correct piezo ratio. We did it symbolically so that we can plot it in a couple of sections to make sure that our ratio isn't extremely sensitive to any of the variables we may not have measured very well.

Data about setup

```
(*wavelength*)
λ = 407.771 * 10^-9;
(*grating spacing*)
d = .001 / 3600;
(*the angle between the normal  of the grating
  and the laser that satisfies the diffraction condition*)
gratingθ = ArcSin[λ/(2 d)];
(*the angle between the grating and the plane of the piezos-0 if mounted  flat,
whatever angle if mounted  otherwise*)
θ1 = gratingθ;
(*the angle between the laser and the plane of the piezo actuators
  (e.g. for Chris' setup, it's 90 degrees-grating angle. For my  setup,
    it's π/2 radians, in general, it will be )*)
θL = π / 2;
(*When you're piezos are at rest,
this represents how far the point you hit the laser at is from  where the vector a
  points. Sort of how far off center you are measured  along the grating.*)
ξ0 = 0;
(*distance between piezo actuators*)
q = (.05972 + .03056) / 2;
(*Thickness-distance from  plane of piezo to plane of grating going
    perpendicularly from  the midpoint  between piezo actuators*)
a = .025;
(*cavity length*)
L = .04;
```

Let's check to make sure that when we expand aBigConstant using our newly found ratio, we get what we think we should.

To first order, aBigConstant should not vary with x1.

In[31]:=
```
(*This is the ratio we think we should get for the setup described above*)
myRATIO  = RATIO[λ, a, q, θ1, d, L, ξ0]
```

Out[31]= 0.478755

In[32]:=
```
test1 = aBigConstant//. myRATIO1  -> myRATIO ;
Series[test1, {x1, 0, 4}]
```

Out[33]= $-1.69407\times10^{-21}\,x1-0.0000294958\,x1^{2}+0.000751895\,x1^{3}-0.0136018\,x1^{4}+O[x1]^{5}$

Nice! It looks like the x1 coefficient is pretty close to zero, probably to within the working precision of the machine.

How RATIO changes as a function of the geometry

We will plot the correct value of RATIO for various likely experimental parameters. Some of the geometry cannot be measured with perfect accuracy, so we mostly want to make sure that the RATIO does not change in some radical way over the values we are likely to have. One example of something we would worry about is if the RATIO became negative for geometrical configurations close to ours since this would require modification of the circuitry that drives the piezo controls.

First, as a function of ax (the distance between the plane of the piezo actuators and the point where the laser strikes the plane of the grating)

In[34]:=
```
Plot[RATIO[λ, ax, q, θ1, d, L, ξ0], {ax, .002, .08}]
```

Out[34]=



Length of the cavity (we can only measure the length of our external cavity to within a few mm at best):

In[35]:=
```
Plot[RATIO[λ, a, q, θ1, d, Lx, ξ0], {Lx, .01, .08}]
```

Out[35]=



We assume that the laser beam would pass through the plane of the piezo actuators exactly in the middle. In this graph, we plot how much the ratio varies depending on how far off center the laser is.

In[36]:=
```
Plot[RATIO[λ, a, q, θ1, d, L, ξx], {ξx, -.01, .01}]
```

Out[36]=



This shows the answer as a function of the actual distance between the actuators.

In[37]:=
```
Plot[RATIO[λ, a, qx, θ1, d, L, ξ0], {qx, .04, .06}]
```

Out[37]=



Check your answer graphically

Blue=laser beam
Green=vector of length αFUNCT that should represent the change in length of the optical cavity
Brown=parts of the piezo mount.
Red=ξ (points along grating surface to show you where things go)
Yellow=a

In[54]:=
```
makeadiagram [x1_, x2_] := Graphics[{
    Arrow[{{0, 0}, {0, Py[x1, x2]}}],
    Text["P", {.001, .3 * Py[x1, x2]}], Thick, Yellow,
    Arrow[{{0, Py[x1, x2]}, {ax[x1, x2], ay[x1, x2] + Py[x1, x2]}}], Thick, Black,
    Text["a", {.5 * ax[x1, x2] + .001, .5 * ay[x1, x2] + Py[x1, x2]}], Thick, Red,
    Line[{{ax[x1, x2], ay[x1, x2] + Py[x1, x2]}, {ax[x1, x2] + ξx[x1, x2, ξFUNCT[x1, x2]],
          ay[x1, x2] + Py[x1, x2] + ξy[x1, x2, ξFUNCT[x1, x2]]}}],
    Text["ξ", {ax[x1, x2] + .5 * ξx[x1, x2, ξFUNCT[x1, x2]],
        ay[x1, x2] + Py[x1, x2] + .5 * ξy[x1, x2, ξFUNCT[x1, x2]] + .002}], Thin, Blue,
    Arrow[{{Lx[0, 0, ξ0] + .02 Cos[θL], Ly[0, 0, ξ0] + .02 Sin[θL]},
        {Lx[0, 0, ξ0] - .04 Cos[θL], Ly[0, 0, ξ0] - .04 Sin[θL]}}], Thick, Green,
    Line[{{Lx[0, 0, ξ0], Ly[0, 0, ξ0]}, {Lx[0, 0, ξ0] + αFUNCT[x1, x2] * Cos[θL],
          Ly[0, 0, ξ0] + αFUNCT[x1, x2] * Sin[θL]}}],
    Text["length=α \n direction of laser",
      {Lx[0, 0, ξ0] + .5 * αFUNCT[x1, x2] * Cos[θL] + .007,
        Ly[0, 0, ξ0] + .5 * αFUNCT[x1, x2] * Sin[θL]}],
    Thin, Brown,
    Circle[{q / 2, 0}, .001],
    Circle[{-q / 2, 0}, .001],
    Text["x1", {-q / 2 + .002, x1 / 2}],
    Text["x2", {q / 2 - .002, x2 / 2}],
    Arrow[{{q / 2, 0}, {q / 2, x2}}],
    Arrow[{{-q / 2, 0}, {-q / 2, x1}}], Thin, Black, Opacity[.8],
    Line[{{Lx[x1, x2, -.01], Ly[x1, x2, -.01]}, {Lx[x1, x2, .01], Ly[x1, x2, .01]}}],
    Opacity[.1],
    Polygon[{{.2 * ax[x1, x2] + q / 2, .2 * ay[x1, x2] + x2}, {q / 2, x2},
        {-q / 2, x1}, {.2 * ax[x1, x2] - q / 2, .2 * ay[x1, x2] + x1},
        {Lx[x1, x2, -.015], Ly[x1, x2, -.015]}, {ax[x1, x2], ay[x1, x2] + Py[x1, x2]},
        {ax[x1, x2], ay[x1, x2] + Py[x1, x2]}, {Lx[x1, x2, .015], Ly[x1, x2, .015]}}],
    Text["Cavity is " <> ToString[αFUNCT[x1, x2] * 10 ^ 6] <> "μm  shorter", {0, -.005}],
    Text["Δλ from  Angle is " <> ToString[-αFUNCT[x1, x2] * λ / L * 10 ^ 9] <> "nm ",
      {0, -.007}], Text["Δλ from  ΔL is " <>
        ToString[(2 d Sin[θ1 + Δθ[x1, x2]] - 2 d Sin[θ1]) * 10 ^ 9] <> "nm ", {0, -.009}]
  }]
```

In[63]:=
```
myX1  = 0
makeadiagram  [myX1 , myX1 *myRATIO ]
myX1  = 0.005
makeadiagram  [myX1 , myX1 *myRATIO ]
myX1  = 0.01
makeadiagram  [myX1 , myX1 *myRATIO ]
myX1  = 0.015
makeadiagram  [myX1 , myX1 *myRATIO ]
```

Out[63]= 0

Out[64]=



Out[65]= 0.005

Out[66]=



Out[67]= 0.01

Out[68]=



Out[69]= 0.015

Out[70]=


ξ
length=α
directionof laser
a
x_1
P
x_2
Cavityis 7434.58 μm shorter
Δλ from Angleis –75.7901 nm
Δλ from ΔL is –71.1294 nm